

Dr Agnieszka Skulimowska

Uniwersytet Przyrodniczo-Humanistyczny w Siedlcach
Wydział Nauk Ścisłych i Przyrodniczych

Dr inż. Krystian Tarczyński

Uniwersytet Przyrodniczo-Humanistyczny w Siedlcach
Wydział Agrobioinżynierii i Nauk o Zwierzętach

VIII. WYBRANE FUNKCJE ARKUSZA KALKULACYJNEGO

WSTĘP

Funkcje pełnią w Excelu kluczową rolę, ich znajomość w powiązaniu z umiejętnością tworzenia formuł, które wykonują obliczenia i generują wyniki, pozwala w nieporównanie większym stopniu korzystać z potencjału programu i oszczędzać czas. W MS Office 2019 są one pogrupowane w kategorie: Finansowe, Logiczne, Tekstowe, Data i godzina, Wyszuk. i odwoł., Matemat. i tryg., Statystyczne, Inżynierskie, Modułowe, Informacyjne, Zgodność, Sieci Web. Niniejszy rozdział przedstawia od podstaw użyteczne funkcje z trzech wymienionych grup: matematycznych, logicznych i statystycznych oraz praktyczne przykłady korzystania z nich.

1. FUNKCJE MATEMATYCZNE

Kategoria zawiera wiele różnorodnych funkcji mniej lub bardziej popularnych. W niektórych sytuacjach można je zastąpić operatorami arytmetycznymi, dotyczy to np. sumy, iloczynu czy potęgi.


1.1. SUMY CZĘŚCIOWE

Pozwalają wykonać jedenaście różnych obliczeń, bo z tylu funkcji można korzystać w ramach tego narzędzia. Jest przydatne, gdy trzeba podsumować pewne grupy danych w tabeli, działa poprawnie dla posortowanych informacji. Sumy częściowe warto wybrać z karty **Dane** (grupa **Konspekt**) i w oknie dialogowym sprecyzować wymagania (il. 1).

Przykład 1. Użyj sum częściowych i policz (plik **1-3.xlsx**, arkusz **Sumy.cz**), ile sprzedano kilogramów owoców a ile warzyw? (**Sumy.cz_roz**)¹.

1. W arkuszu **Sumy.cz** posortuj od **A do Z** lub **od Z do A** dane w kolumnie **A** (**Dane > Sortowanie i filtrowanie**).
2. Na karcie **Dane** w grupie **Konspekt** wybierz **Suma częściowa**.
3. W oknie dialogowym: **Dla każdej zmiany w** wskaż **Rodzaj**, potem **Użyj funkcji Suma**, a dla **Dodaj sumę częściową do** zaznacz **Sprzedaż w kg** i kliknij **OK** (il. 1).

¹ Nazwa arkusza z rozwiązaniem przykładem.



1. Okno dialogowe *Sumy częściowe*

4. Uaktywnij komórkę **C20** i przyjrzyj się formule **=SUMY.CZĘŚCIOWE(9;C2:C19)** (il. 2). Podobnie sprawdź zawartość komórki **C44**. W obu przypadkach pierwszy argument to liczba 9, która informuje o użyciu funkcji **SUMA**, drugi dotyczy zakresu komórek.

C20

:

✕

✓

f_x

=SUMY.CZĘŚCIOWE(9;C2:C19)

1	2	3		A	B	C	D
	1			Rodzaj	Nazwa	Sprzedaż w kg	Miejsce
	20			owoc Suma		455	
	44			warzywo Suma		796	
	45			Suma końcowa		1251	

2. Wynik funkcji w przykładzie 1

5. Sprzedano 455 kg owoców i 796 kg warzyw (il. 2).

Użycie sum częściowych powoduje wstawianie dodatkowych wierszy oraz tworzenie **konspetu** arkusza, który umożliwia chowanie i odkrywanie szczegółowych informacji w tabeli (il. 2).

1.2. SUMA JEŻELI

Funkcja dodaje z określonego zakresu te wartości, które spełniają ustalone kryterium. Ma następującą składnię:

SUMA.JEŻELI(zakres;kryteria;[suma_zakres])²

Przy czym:

- 1) **zakres** - jest argumentem koniecznym, odnosi się do komórek, w których obowiązuje podane kryterium. Mogą one zawierać liczby, odwołania do nich, nazwy, tablice, daty określonego formatu, pomijane są natomiast wartości puste i tekstowe;
- 2) **kryteria** - argument obowiązkowy, wskazuje komórki do sumowania, przyjmuje różne formy: liczb, wyrażeń, odwołań do komórek, tekstu czy funkcji. Dozwolone jest korzystanie z pytajnika (?) zastępującego pojedynczy znak czy gwiazdki (*) reprezentującej dowolny ich ciąg. Natomiast podczas szukania * lub ? trzeba te symbole poprzedzić znakiem tyldy (~). Poza liczbami wszystkie kryteria potrzebują cudzysłowów;
- 3) **suma_zakres** - argument opcjonalny³, wskazuje komórki do sumowania tylko wtedy, gdy są one inne niż te podane przez argument **zakres**.

² Na końcu składni funkcji i list z formułami nie ma znaków interpunkcyjnych, aby zapobiec potencjalnym błędom.

³ W składni funkcji argumenty ujęte w nawiasy kwadratowe [] nie są obowiązkowe.

Przykład 2. Skorzystaj z funkcji **suma.jeżeli** (plik **1-3.xlsx**, arkusz **Suma.jeż**), aby policzyć ile sprzedano kilogramów: owoców, warzyw, owoców i warzyw, których nazwy zaczynają się na literę **m**? (**Suma.jeż_roz**).

Do komórek **F43**, **F44**, **F45** (il. 3) wpisz formuły:

- 1) **=SUMA.JEŻELI(A2:A42;"owoc";C2:C42)**
- 2) **=SUMA.JEŻELI(A2:A42;"warzywo";C2:C42)**
- 3) **=SUMA.JEŻELI(B2:B42;"m*";C2:C9)**

Odpowiedzi na pytania przedstawia il. 3.

	C	D	E	F
43			Ile sprzedano kilogramów owoców?	455
44			Ile sprzedano kilogramów warzyw?	796
45			Ile sprzedano kilogramów owoców i warzyw, których nazwy zaczynają się na literę m?	194

3. Liczby kilogramów sprzedanych owoców i warzyw

1.3. SUMA WARUNKÓW

Działa podobnie jak **suma.jeżeli**, ale może zawierać więcej niż jedno kryterium. Funkcja sumuje wartości spełniające określone warunki i ma postać:

SUMA.WARUNKÓW(suma_zakres;kryteria_zakres1;kryteria1;kryteria_zakres2;kryteria2;...)

Natomiast:

- 1) **suma_zakres** - to argument obowiązkowy, wskazuje komórki do sumowania;
- 2) **kryteria_zakres1** - argument konieczny, jest zakresem kontrolowanym przez **kryteria1**;
- 3) **kryteria1** - argument wymagany, potrzebny do wskazania, które komórki z obszaru **kryteria_zakres1** należy dodać;
- 4) **kryteria_zakres2;kryteria2** – argumenty niekonieczne, kolejny zakres i powiązane z nim kryterium.

Przykład 3. Wykorzystaj adresowanie względne i w kolumnie **F** (plik **1-3.xlsx**, arkusz **Suma.war**) policz, ile kg owoców i warzyw sprzedano na każdym ze stoisk? Te same obliczenia wykonaj w kolumnie **G**, ale z myślą o dogodnym kopiowaniu formuł zastosuj adresowanie mieszane (**Suma.warunków_roz**).

W arkuszu **Suma.warunków** wpisz formuły do odpowiednich komórek kolumny **F**:

- 1) **=SUMA.WARUNKÓW(C2:C42;A2:A42;"owoc";D2:D42;"Stoisko 1")**
- 2) **=SUMA.WARUNKÓW(C2:C42;A2:A42;"owoc";D2:D42;"Stoisko 2")**
- 3) **=SUMA.WARUNKÓW(C2:C42;A2:A42;"owoc";D2:D42;"Stoisko 3")**
- 4) **=SUMA.WARUNKÓW(C2:C42;A2:A42;"warzywo";D2:D42;"Stoisko 1")**
- 5) **=SUMA.WARUNKÓW(C2:C42;A2:A42;"warzywo";D2:D42;"Stoisko 2")**
- 6) **=SUMA.WARUNKÓW(C2:C42;A2:A42;"warzywo";D2:D42;"Stoisko 3")**

Wykorzystaj adresowanie mieszane w kolumnie **G**:

- 1) **=SUMA.WARUNKÓW(C\$2:C\$42;A\$2:A\$42;"owoc";D\$2:D\$42;"Stoisko 1")**
- 2) **=SUMA.WARUNKÓW(C\$2:C\$42;A\$2:A\$42;"owoc";D\$2:D\$42;"Stoisko 2")**
- 3) **=SUMA.WARUNKÓW(C\$2:C\$42;A\$2:A\$42;"owoc";D\$2:D\$42;"Stoisko 3")**
- 4) **=SUMA.WARUNKÓW(C\$2:C\$42;A\$2:A\$42;"warzywo";D\$2:D\$42;"Stoisko 1")**
- 5) **=SUMA.WARUNKÓW(C\$2:C\$42;A\$2:A\$42;"warzywo";D\$2:D\$42;"Stoisko 2")**
- 6) **=SUMA.WARUNKÓW(C\$2:C\$42;A\$2:A\$42;"warzywo";D\$2:D\$42;"Stoisko 3")**

Liczby kilogramów sprzedanych owoców i warzyw przez każde ze stoisk handlowych przedstawia il. 4.

	E	F	G
43		Adresowanie względne	Adresowanie mieszane
44	Ile kilogramów owoców sprzedano na Stoisku 1?	186	186
45	Ile kilogramów owoców sprzedano na Stoisku 2?	194	194
46	Ile kilogramów owoców sprzedano na Stoisku 3?	75	75
47	Ile kilogramów warzyw sprzedano na Stoisku 1?	292	292
48	Ile kilogramów warzyw sprzedano na Stoisku 2?	287	287
49	Ile kilogramów warzyw sprzedano na Stoisku 3?	217	217

4. Sprzedane owoce i warzywa przez każde ze stoisk

2. FUNKCJE LOGICZNE

Wykorzystuje się je do sprawdzania warunków w komórkach wypełnionych danymi. Występują samodzielnie lub jako argumenty innych funkcji.

2.1. JEŻELI

Jest jedną z najpopularniejszych funkcji w Excelu, kontroluje warunek logiczny i przekazuje jedną wartość, gdy prawda a inną, kiedy fałsz. Ma postać:


JEŻELI(test_logiczny;[wartość_jeżeli_prawda];[wartość_jeżeli_fałsz])

W tym:

- 1) **test_logiczny** - jest argumentem obowiązkowym, może być dowolnym wyrażeniem, wartością, funkcją, której wynikiem jest PRAWDA lub FAŁSZ, w szczególności też liczbą lub formułą, która ją generuje;
- 2) **wartość_jeżeli_prawda** - argument opcjonalny, który wyświetla się, gdy **test_logiczny** jest PRAWDA lub liczba jest inna niż zero;
- 3) **wartość_jeżeli_fałsz** - argument niekonieczny, który wyświetla się, gdy **test_logiczny** jest FAŁSZEM lub liczba jest równa zero.

Przykład 4. Oblicz prowizję (plik **4-8.xlsx**, arkusz **Jeżeli**), która dla sprzedaży wynoszącej 20 000 zł i więcej wynosi 12%, w pozostałych przypadkach funkcja **jeżeli** ma zwracać tekst **Nie** (**Jeżeli_roz**).

1. Do komórki **C2** arkusza **Jeżeli** wprowadź formułę **=JEŻELI(B2>=20000;B2*12%;"Nie")** i przeanalizuj jej działanie (il. 5).



5. Wynik działania funkcji jeżeli w przykładzie 4

2. Użyj uchwytu wypełniania i przeciągnij formułę w dół aż do komórki **C24** (il. 6).

	A	B	C
1	Handlowiec	Sprzedaż	Prowizja
2	Abacki Karol	10299	Nie
3	Babacka Anna	15085	Nie
4	Cabacki Janusz	11782	Nie
5	Dabacka Olga	15765	Nie
6	Ebacki Krzysztof	14380	Nie
7	Fabacka Ewa	16337	Nie
8	Gabacka Janina	16704	Nie
9	Habacka Ewa	18514	Nie
10	Ibacka Anna	12440	Nie
11	Kabacki Marian	20156	2419
12	Labacka Danuta	20734	2488
13	Łabacka Wiesława	19775	Nie
14	Mabacka Iwona	20108	2413
15	Nabacki Jan	18979	Nie
16	Obacka Ewa	14628	Nie
17	Pabacki Stanisław	19103	Nie
18	Rabacki Jakub	20417	2450
19	Sabacka Urszula	12060	Nie
20	Tabacka Małgorzata	18194	Nie
21	Ubacki Wojciech	13685	Nie
22	Wabacki Roman	9116	Nie
23	Zabacka Jolanta	15387	Nie
24	Żabacka Bożena	11051	Nie

6. Wynik formuły dla przykładu 4

Przykład 5. Spowoduj, aby funkcja **jeżeli** generowała wartość PRAWDA (plik 4-8, arkusz **Jeżeli2**), gdy w kolumnie A liczba będzie różna od zera, a FAŁSZ jeśli równa zero (**Jeżeli2_roz**).

1. Do komórki **B2** wpisz formułę **=JEŻELI(A2;PRAWDA;FAŁSZ)**
2. Skopiuj ją do pozostałych komórek kolumny **B** (il. 7).

	A	B
1	Wartość	Liczba jako test logiczny
2	-15	PRAWDA
3	43	PRAWDA
4	37	PRAWDA
5	-1	PRAWDA
6	-15	PRAWDA
7	27	PRAWDA
8	44	PRAWDA
9	0	FAŁSZ
10	-3	PRAWDA
11	-3	PRAWDA
12	-13	PRAWDA
13	15	PRAWDA
14	0	FAŁSZ
15	22	PRAWDA
16	34	PRAWDA
17	13	PRAWDA
18	25	PRAWDA
19	11	PRAWDA
20	0	FAŁSZ

7. Wynik działania funkcji *jeżeli* dla przykładu 5

2.2. LUB

Funkcja przekazuje wartość PRAWDA, gdy chociaż jeden z argumentów jest zgodny z rzeczywistością i FAŁSZ w przeciwnym razie. Ma następującą składnię:

LUB(warunek_logiczny1;[warunek_logiczny2];...)

Argumentów może być do 255, przy czym **warunek_logiczny1** jest obowiązkowy a pozostałe są niekonieczne.

Przykład 6. Firma szuka pracowników, do jednego działu posiadających prawo jazdy kategorii B, a do drugiego znających język angielski na poziomie C1 (plik **4-8.xlsx**, arkusz **Lub**). Spraw, aby przy kandydatach spełniających jedno z kryteriów wyświetlała się wartość PRAWDA, a obok pozostałych FAŁSZ (**Lub_roz**).

1. Do komórki **B4** wprowadź formułę **=LUB(B2=1;B3=3)** i skopiuj ją do pozostałych kandydatów (il. 8).

	A	B	C	D	E	F	G	H
1	Kryteria	Miłosz Zgódka	Arkadiusz Niewiadomski	Zygmunt Zadrozny	Piotr Lesiński	Marta Kuźma	Witold Pietraszak	Andrzej Jówko
2	Prawo jazdy kategorii B (0-nie; 1-tak)	1	0	0	1	1	1	0
3	Znajomość języka angielskiego (1-B1; 2-B2; 3-C1)	2	1	2	3	2	1	1
4		PRAWDA	FAŁSZ	FAŁSZ	PRAWDA	PRAWDA	PRAWDA	FAŁSZ

8. Wynik działania funkcji *oraz* w przykładzie 5

2. Zwróć uwagę przy których osobach wyświetla się wartość PRAWDA (il. 8).

2.3. ORAZ

Funkcja testuje warunki i zwraca wartość PRAWDA, gdy wszystkie są prawdziwe i FAŁSZ, jeśli chociaż jeden jest fałszywy. Jej składnia jest następująca:

ORAZ(warunek_logiczny1;[warunek_logiczny2];...)

Przy czym **warunek_logiczny1** to argument konieczny, pozostałe są nieobowiązkowe, każdy może być wartością logiczną, tablicą, odwołaniem, musi zwracać PRAWDE lub FAŁSZ, a ich liczba nie może przekroczyć 255.

Przykład 7. Kandydat do pracy (plik **4-8.xlsx**, arkusz **Oraz**) ma spełnić kilka warunków:

1. Posiadać egzamin maturalny z matematyki zdany na poziomie rozszerzonym - minimum 90%.
2. Mieć prawo jazdy kategorii B.
3. Znać język angielski na poziomie C1.

Skorzystaj z funkcji **oraz**, aby przy osobach spełniających wszystkie kryteria wyświetlała się wartość PRAWDA, a przy pozostałych FAŁSZ (**Oraz_roz**).

1. Do komórki **B5** arkusza **Oraz** wprowadź formułę **=ORAZ(B2>=90;B3="Tak";B4=3)**
2. Przekopiu ją, aby pojawiły się wyniki przy pozostałych kandydatach (il. 9).

	A	B	C	D	E	F	G	H
1	Kryteria	Miłosz Zgódka	Arkadiusz Niewiadomsk	Zygmunt Zadroźn	Piotr Lesiński	Marta Kuźma	Witold Pietraszak	Andrzej Jówko
2	Egzamin maturalny z matematyki na poziomie rozszerzonym w %	66	92	97	91	45	85	75
3	Prawo jazdy	Tak	Tak	Nie	Tak	Tak	Tak	Nie
4	Znajomość języka angielskiego (1-B1; 2-B2; 3-C1)	1	1	3	3	2	1	1
5		FAŁSZ	FAŁSZ	FAŁSZ	PRAWDA	FAŁSZ	FAŁSZ	FAŁSZ

9. Rezultat działania funkcji **oraz**

Przykład 8. Przygotuj formułę, aby przy każdej osobie, która pomyślnie zaliczyła testy sprawnościowe i z angielskiego bądź francuskiego uzyskała minimum 20 punktów, pojawił się komunikat **Kolejny etap** (plik **4-8.xlsx**, arkusz **JeżeliOrazLub**), a przy pozostałych kandydatach nie wyświetlało się nic (**JeżeliOrazLub_roz**).

1. Do komórki **E2** wprowadź formułę:
=JEŻELI(ORAZ(B2="Tak";LUB(C2>=20;D2>=20));"Kolejny etap";"")
2. Skopiu ją do pozostałych komórek w kolumnie **E** (il. 10).

	A	B	C	D	E
1	Imię i nazwisko	Testy sprawnościowe	Język angielski	Język francuski	Komunikat
2	Andrzej Janik	Tak	19	17	
3	Zygmunt Kałużny	Nie	23	19	
4	Piotr Lasek	Tak	18	18	
5	Ewa Maczek	Tak	22	22	Kolejny etap
6	Jolanta Manik	Nie	25	25	
7	Jan Niwiński	Nie	18	15	
8	Anna Nowacka	Nie	19	20	
9	Witold Płatek	Tak	23	21	Kolejny etap
10	Marek Zgoliński	Nie	24	17	
11	Piotr Żatek	Tak	18	19	

10. Wynik formuły w przykładzie 8

3. FUNKCJE STATYSTYCZNE

Statystyka jest nauką, która zajmuje się metodami pozyskiwania, przedstawiania, a głównie analizowania danych charakterystycznych dla zjawisk masowych⁴. Typowa dla nich jest duża liczba danych a przykładami są wyniki egzaminów maturalnych, daty urodzenia, poparcie w wyborach partii politycznych czy stany magazynowe.

3.1. MINIMUM I MAXIMUM

Funkcje działają tak, że MIN zwraca najmniejszą liczbę ze zbioru wartości, a MAX największą. Ich składnie są następujące:

MIN(liczba1;[liczba2];...)

MAX(liczba1;[liczba2];...)

Każda z funkcji może mieć do 255 argumentów w postaci liczb, odwołań do nich, nazw czy tablic, ale obowiązkowa jest tylko **liczba1**.

3.2. ŚREDNIA

To podstawowa i prosta w interpretacji funkcja statystyczna, która charakteryzuje zbiór poprzez obliczenie średniej arytmetycznej podanych liczb:

ŚREDNIA(liczba1;[liczba2];...)

Dozwolonych jest do 255 argumentów, które mogą być liczbami, nazwami i odwołaniami do komórek lub zakresów, a konieczna jest tylko **liczba1**.

Przykład 9. Oblicz **min**, **max** oraz **średnią** z dokładnością do dwóch miejsc po przecinku (plik **9-14.xlsx**, arkusz **MinMaxŚrednia**). Czy funkcje biorą pod uwagę komórki puste? (**Min-MaxŚrednia_roz**).

Do komórek z zakresu **B14:B16** (il. 11) wpisz funkcje:

1) **=MIN(B2:B13)**

2) **=MAX(B2:B13)**

3) **=ŚREDNIA(B2:B13)**

Ustal dokładność wyświetlania wartości średniej w komórce **B16** (**Narzędzia główne > Liczba**). Czy pusta komórka **B3** ma wpływ na funkcje?

⁴ https://stat.gov.pl/cps/rde/xbcr/rzesz/ASSETS_skrypt.pdf; data dostępu: 15.10.2023.

	A	B
1	Miesiąc	Kwota brutto [zł]
2	styczeń	4500
3	luty	
4	marzec	6200
5	kwiecień	6400
6	maj	7600
7	czerwiec	8000
8	lipiec	7400
9	sierpień	7000
10	wrzesień	6400
11	październik	7200
12	listopad	7400
13	grudzień	4700
14	Min	4500
15	Max	8000
16	Średnia	6618,18

11. Wartości działania funkcji w przykładzie 9

3.3. LICZ JEŻELI

Funkcja zlicza komórki, które spełniają podany warunek i ma postać:

LICZ.JEŻELI(zakres;kryteria)

Przy czym:

- 1) **zakres** to argument konieczny, informuje, które komórki mają być sprawdzane. Może być liczbą, odwołaniem do niej, nazwanym zakresem, tablicą.
- 2) **kryteria** są pojedynczym warunkiem determinującym, które komórki mają być liczone.

Przykład 10. W arkuszu **Licz.jeżeli** (plik **9-14.xlsx**) policz, ile osób:

- 1) urodziło się w Polsce?
- 2) ma ponad 55 lat?
- 3) urodziło się po 1970 roku?
- 4) ma imiona, które składają się z 3 liter?
- 5) urodziło się w miejscowości, której nazwa zaczyna się na literę **W**⁵? (**Licz.jeżeli_roz**).

Do komórek od **M3** po **M7** (il. 12) wpisz kolejno formuły:

- 1) **=LICZ.JEŻELI(H2:H24;"Polska")**

Możesz również odwołać się do adresu **H2**, gdzie jest tekst **Polska**, wtedy formuła będzie następująca **=LICZ.JEŻELI(H2:H24;H2)**

- 2) **=LICZ.JEŻELI(K2:K24;">55")**
- 3) **=LICZ.JEŻELI(F2:F24;">31.12.1970")**
- 4) **=LICZ.JEŻELI(B2:B24;"??")**
- 5) **=LICZ.JEŻELI(G2:G24;"W*")**

⁵ A. Skulimowska, *Technologia informacyjna. Excel*, Wydawnictwo UPH, Siedlce 2017, s. 185-186.

	L	M
2	Ile osób:	
3	urodziło się w Polsce?	11
4	ma ponad 55 lat?	5
5	urodziło się po 1970 roku?	14
6	ma imiona, które składają się z 3 liter?	4
7	urodziło się w miejscowości, której nazwa zaczyna się na literę W?	6

12. Wyniki do przykładu 10

3.4. LICZ WARUNKI

Funkcja sprawdza wiele warunków w różnych zakresach i zlicza komórki, które je spełniają. Ma następującą składnię:

LICZ.WARUNKI(zakres_kryterium1;kryterium1;[zakres_kryterium2;kryterium2]...)

Natomiast:

- 1) **zakres_kryterium1** - to argument obowiązkowy, informuje w którym obszarze sprawdzane jest **kryterium1**;
- 2) **kryterium1** - jest argumentem koniecznym i przedstawia pierwszy sprawdzany warunek. Może być adresem komórki, liczbą, wyrażeniem lub tekstem w cudzysłowie;
- 3) **zakres_kryterium2;kryterium2** - argumenty opcjonalne, powiązane ze sobą zakresy i testowane w nich kryteria. Łącznie takich par zakres-kryterium może być do 127.

Przykład 11. W arkuszu **Licz.warunki** (plik 9-14.xlsx) policz, ile kobiet urodziło się:

- 1) po 1965 roku?
- 2) po 1965 roku i poza Polską?
- 3) po 1965 roku, poza Polską i nosi nazwisko zaczynające się na literę **L**? (**Licz.warunki_roz**).

Do komórek **M3**, **M4**, **M5** arkusza **Licz.warunki** wpisz kolejno formuły:

- 1) **=LICZ.WARUNKI(J2:J24;"K";F2:F24;">31.12.1965")**
- 2) **=LICZ.WARUNKI(J2:J24;"K";F2:F24;">31.12.1965";H2:H24;"<>Polska")**
- 3) **=LICZ.WARUNKI(J2:J24;"K";F2:F24;">31.12.1965";H2:H24;"<>Polska";A2:A24;"L*")**

Przeanalizuj je i porównaj otrzymane liczby (il. 13).

	L	M
2	Przykład 10. Policz ile kobiet urodziło się:	
3	1) po 1965 roku?	10
4	2) po 1965 roku i poza Polską?	6
5	2) po 1965 roku, poza Polską i nosi nazwisko zaczynające się na literę L ?	1

13. Wyniki formuł w przykładzie 11

3.5. MEDIANA

Nazywana jest wartością środkową uporządkowanego zbioru. Jeśli dane są ułożone rosnąco a liczba elementów jest nieparzysta, to medianą **jest liczba środkowa**, a w przypadku parzystej - **średnia arytmetyczna dwóch liczb środkowych**.

Funkcja ma następującą składnię:

MEDIANA(liczba1;[liczba2];...)

Argumentem koniecznym jest **liczba1**, a **liczba2** i kolejne są nieobowiązkowe. Łącznie funkcja może mieć od 1 do 255 argumentów w postaci liczb lub odwołań do nich, nazw bądź tablic.

Przykład 12. Dla zestawów liczb w kolumnach **B** i **D** (arkusz **Mediana**, plik **9-14.xlsx**) wyznacz medianę bez korzystania z Excela, a w kolumnach **C** i **E** z funkcją **mediana**. Zinterpretuj otrzymany wynik (**Mediana_roz**).

1. Uporządkuj rosnąco liczby w kolumnie **B**.
2. Liczba elementów jest parzysta, średnia arytmetyczna dwóch liczb środkowych (2 oraz 3), to **2,5** i tyle wynosi **mediana**.
3. Uszereguj rosnąco liczby w kolumnie **D**.
4. Liczba danych jest nieparzysta, dlatego **mediana** równa się wartości środkowej i wynosi **7200** (il. 14).

	A	B	C	D	E
1		-7	-4		
2		-4	5	6400	7200
3		-3	0	6400	6400
4		0	4	7000	7600
5		2	-3	7200	8000
6		3	2	7200	7400
7		4	4	7200	7000
8		4	7	7400	6400
9		5	-7	7600	7200
10		7	3	8000	7200
11	Mediana	2,5	2,5	7200	7200

14. Mediana w przykładzie 12

5. Do komórek **C11** i **E11** wpisz odpowiednio formuły **=MEDIANA(C1:C10)** oraz **=MEDIANA(E2:E10)** (il. 14).
6. Dla zbiorów w kolumnach:
 - 6.1. **B** oraz **C mediana** wynosi **2,5** i informuje, że połowa liczb jest mniejsza od tej wartości a druga połowa większa od niej.
 - 6.2. **D i E mediana** równa się **7200**, co świadczy, że 50% liczb ma wartości niższe bądź równe medianie, a drugie 50 % równe jej lub wyższe.

3.6. ODCHYLENIE STANDARDOWE I WARIANCJA

Odchylenie standardowe jest miarą rozbieżności wartości zmiennej od średniej arytmetycznej. Jeśli zróżnicowanie jest małe to i wartość relatywnie niewielka, duża dotyczy wysokiego rozproszenia. O zmienności wyników świadczy **wariancja** jako kwadrat odchylenia standardowego.

Obie funkcje mają następujące składnie:

ODCH.STAND.POPUL(liczba1;[liczba2];...)

WARIANCJA.POPUL(liczba1;[liczba2];...)

Argumenty dotyczą całej populacji i tylko **liczba1** jest obowiązkowa, pozostałe są opcjonalne. Lista może liczyć od 1 do 255 elementów i zawierać: liczby, odwołania do nich, nazwy i tablice. Uwzględniane są również bezpośrednio wprowadzone wartości logiczne i tekstowe reprezentacje liczb, a pomijane, gdy znajdują się w odwołaniach jako części obliczeń⁶. W przypadku tablic, ale też odwołań zliczane są wyłącznie znajdujące się w nich liczby, a ignorowane teksty, komórki puste, wartości błędów oraz logiczne. Odchylenie standardowe w populacji można także szacować na podstawie próby, która ją reprezentuje, służą do tego funkcje **ODCH.STANDARD.PRÓBK**i i uwzględniająca tekst oraz wartości logiczne **ODCH.STANDARDOWE.A**, analogicznie jest dla wariancji - **WARIANCJA.PRÓBK**i i **WARIANCJA.A**.

⁶ <https://support.microsoft.com/pl-pl/office/odch-stand-popul-funkcja-6e917c05-31a0-496f-ade7-4f4e7462f285>, data dostępu: 12.10.2023.

Przykład 13. Arkusz **Odch.stand** (plik **9-14.xlsx**) zawiera miesięczne wynagrodzenia wszystkich zatrudnionych osób w firmie. Z dokładnością do jedności oblicz średnią arytmetyczną kwot do wypłaty oraz wartość odchylenia standardowego i zinterpretuj ją? (**Odch.stand_roz**).

Wpisz formuły (il. 15) do komórek **C25** i **C26**:

- 1) =ŚREDNIA(C2:C24)
- 2) =ODCH.STAND.POPUL(C2:C24)

Wartość odchylenia standardowego równa **1201** oznacza, że w badanej populacji wynagrodzenia zatrudnionych osób odchylają się przeciętnie od średniej arytmetycznej o plus minus **1201 zł**. Skoro średnie wynagrodzenie wynosi **4750 zł**, to teoretycznie pensje mogą być od tej kwoty o **1201 zł** niższe jak i wyższe.

	A	B	C
20	Tabacka	Małgorzata	6224
21	Ubacki	Wojciech	6710
22	Wabacki	Roman	3255
23	Zabacka	Jolanta	3076
24	Żabacka	Bożena	5634
25	Średnia arytmetyczna		4750
26	Odchylenie standardowe w populacji		1201

15. Wartości średniej arytmetycznej i odchylenia standardowego w przykładzie 14

Przykład 14. W arkuszu **Wariancja** (plik **9-14.xlsx**) znajdują się wyniki pomiaru temperatury ciała wszystkich pacjentów z sal szpitalnych. Oblicz, z dokładnością do części dziesiętnych, średnią arytmetyczną dla każdej z sal i wariancję, aby wiedzieć w której z nich wyniki są najbardziej zróżnicowane? (**Wariancja_roz**).

1. Do komórki **H2** wpisz =ŚREDNIA(B2:G2) i skopiuj formułę do **H3** i **H4** (il. 16).
2. Podobnie do **I2** wprowadź =WARIANCJA.POP(B2:G2) i skopiuj do **I3** oraz **I4** (il. 16).
3. Ustaw dokładność wyświetlania liczb w kolumnach **H** i **I** (Narzędzia główne > Liczba >

Zmniejsz dziesiętne ).

	A	B	C	D	E	F	G	H	I
								Średnia arytmetyczna	Wariancja
1	Pomiar temperatury:								
2	Sala 1	36	38	37	36	37	38	37,0	0,7
3	Sala 2	34	37	39	38	35	39	37,0	3,7
4	Sala 3	37	37	37	37	37	37	37,0	0,0

16. Wartości średniej arytmetycznej i wariancji dla przykładu 14

4. W każdej z sal średnia temperatura jest taka sama i wynosi 37,0. Największa wariancja **3,7** jest dla sali 2 i tu wyniki są najbardziej zróżnicowane, dla sali 1 wynosi ona 0,7, a dla sali 3 jest równa 0, bo każdy pacjent miał taką samą temperaturę.

4. ZADANIA

1. Dodaj wartości poszczególnych sprzedawców za pomocą:
 - 1.1. Sum częściowych (plik **1-7.xlsx**, arkusz **Sumy.cz**).
 - 1.2. Funkcji **suma.jezeli** (plik **1-7.xlsx**, arkusz **Suma.jezeli**).
2. Dla sprzedawców z Siedlec i Mińska Mazowieckiego dodaj wartości, które są **większe od 20000 zł** (plik **1-7.xlsx**, arkusz **Suma.war**).

3. Studenci, którzy uzyskali 51% i więcej z 78 możliwych punktów zdali egzamin. Przygotuj funkcję **jeżeli** tak, aby w kolumnie **D** wyświetlał się tekst **Niezdany** lub **Zdany** (plik **1-7.xlsx**, arkusz **Jeżeli**).
4. Kontrahent powinien dostarczać rocznie **140 części**, w I jak i II półroczu ma realizować dokładnie **50%** zamówienia. Każdy z dostawców oferuje, że jeśli wartość rocznego zamówienia przekroczy **1500 zł** nabywca dostanie rabat w wysokości **100 zł**. Wskaż najtańszego kontrahenta, który spełnia wymagania (plik **1-7.xlsx**, arkusz **JeżeliOraz**).
 - 4.1. W wierszu **Razem** zsumuj liczbę elementów jaką oferuje w I i II półroczu każdy z kontrahentów.
 - 4.2. Na pytanie **Czy dostawca spełnia wymagania?** odpowiedz **PRAWDA** bądź **FAŁSZ**. Warunki dotyczące liczby części w I i II półroczu a także rocznie (**Razem**) porównaj ze stawianymi wymaganiami za pomocą funkcji **Oraz**.
 - 4.3. Dla dostawców, którzy spełniają wymagania (**PRAWDA** w wierszu 5) policz:
 - 4.3.1. **Wartość zamówienia** uwzględniając cenę jednostkową i zapotrzebowanie.
 - 4.3.2. **Wartość zamówienia z rabatem** - skorzystaj z funkcji **jeżeli**.
5. W arkuszu **Jeżeli2** (plik **1-7.xlsx**) policz:
 - 5.1. W kolumnach **F**, **G** i **H** koszty z uwzględnieniem podanych stawek.
 - 5.2. **Oplaty razem** zagnieżdżając funkcję **jeżeli**. Użyj formatowania walutowego i symbolu zł. Rozpatrz trzy przypadki dotyczące wody i ścieków, które występują w kolumnie D:
 - 5.2.1. Dla tekstu **Niedopłata** dodaj kwotę różnicy z rozliczenia.
 - 5.2.2. Jeśli wyświetla się tekst **Nadpłata** odejmij wartość różnicy.
 - 5.2.3. Gdy komórka jest pusta nie uwzględniaj kolumny E w opłatach.
6. Dla danych z zakresu B2:F23 oblicz funkcje: min, max, średnią, medianę, odchylenie standardowe i wariancję (plik **1-7.xlsx**, arkusz **Statystyczne**).
7. Skorzystaj z funkcji **licz.jeżeli** i **licz.warunki** (plik **1-7.xlsx**, arkusz **Licz.jeżeliLicz.warunki**)⁷, aby odpowiedzieć na pytania:
 - 7.1. Ile jest **kobiet**?
 - 7.2. Ile osób urodziło się **poza Polską**?
 - 7.3. Ilu jest mężczyzn, którzy urodzili się **po 1960 roku**?
 - 7.4. Ilu jest mężczyzn, którzy urodzili się **po 1960 roku w Polsce**?
 - 7.5. Ilu jest mężczyzn, którzy urodzili się **po 1960 roku, w Polsce** a nazwisko rodowe ich matki zaczyna się na literę **G**?

LITERATURA

A. Skulimowska, *Technologia informacyjna. Excel*, Wydawnictwo UPH, Siedlce 2017, s. 152-227.
https://stat.gov.pl/cps/rde/xbcr/rzesz/ASSETS_skrypt.pdf; data dostępu: 15.10.2023.
<https://support.microsoft.com/pl-pl/office/odch-stand-popul-funkcja-6e917c05-31a0-496f-ade7-4f4e7462f285>, data dostępu: 12.10.2023.

⁷ A. Skulimowska, *Technologia informacyjna. Excel*, Wydawnictwo UPH, Siedlce 2017, s. 184-188.

SPIS TREŚCI

Wstęp	1
1. Funkcje matematyczne	1
1.1. Sumy częściowe	1
1.2. Suma jeżeli	2
1.3. Suma warunków	3
2. Funkcje logiczne	4
2.1. Jeżeli	4
2.2. Lub	6
2.3. Oraz	7
3. Funkcje statystyczne	8
3.1. Minimum i maximum	8
3.2. Średnia	8
3.3. Licz jeżeli	9
3.4. Licz warunki	10
3.5. Mediana	10
3.6. Odchylenie standardowe i wariancja	11
4. Zadania	12
Literatura	13